Actividad 8: Iniciandose en Computo Simbolico con Maxima

Martin Alejandro Paredes Sosa

Abril, 2016

1. Introducción

Maxima es una herramienta de de cálculo bastante versátil. Es un sistema para la manipulación de expresiones simbólicas y numéricas, incluyendo diferenciación, integración, ecuaciones diferenciales ordinarias, sistema de ecuaciones linales, vectores, matrices, entre otros. Maxima produce resultados de alta precisión. Adicionalmete permirte la graficación de funciones y datos en dos y tres dimensiones. [1]

```
(%i3) integrate ( 1 / (1 + x^4), x);

           2                            2
      log(x  + sqrt(2) x + 1)      log(x  - sqrt(2) x + 1)
(%o3) ----------------------- - -----------------------
            4 sqrt(2)                    4 sqrt(2)
                    2 x + sqrt(2)            2 x - sqrt(2)
               atan(-------------)      atan(-------------)
                       sqrt(2)                  sqrt(2)
             + ------------------- + -------------------
                    2 sqrt(2)                2 sqrt(2)
```

Figura 1: Interfaz de Maxima

Para esta practica se nos pidió familiarizarnos con esta herramienta. Para esto se utilizó el manual de Jay Kerns sobre calculo de varias variables [2], recreando los ejemplos que se mostraban. Para la realización de esta práctica, se utilzó wxMaxima, el cual es una interfaz grafica para trabajar con los comandos de Maxima.

2. Geometria en tres dimensiones

Esta sección consta en enseñarnos herramientas para geometria tridimensional.

2.1. Vectores y Algebra lineal

En maxima hay forma de realizar operaciones con vectores, como es el producto punto y el producto cruz.

```
(%i1)  a: [6,2,5];
       b: [8,-3,0];
       a.b;
       load(vect);
       express(a~b);
       c: [-5,2,9];
       express(a.(b~c));
```

(%o1) $[6, 2, 5]$

(%o2) $[8, -3, 0]$

(%o3) 42

(%o4) $/usr/share/maxima/5.34.1/share/vector/vect.mac$

(%o5) $[15, 40, -34]$

(%o6) $[-5, 2, 9]$

(%o7) -301

2.2. Lineas, Planos y Superficies Cuadraticas

Con maxima se pueden definir ecuaciones de planos y superficies, con el objetivo de poder visualizarlos.

```
(%i1)  load(draw);
       ellips1: x^2/3+0.5*x*y+z   = 0;
       draw3d(enhanced3d = true,
              palette = [cyan,blue,cyan],
              implicit(ellips1, x,-100,100, y,-100,100, z,-100,100));
```

(%o1) $/usr/share/maxima/5.34.1/share/draw/draw.lisp$

(%o2) $z + 0.5\,x\,y + \frac{x^2}{3} = 0$

(%o3) $[\mathrm{gr3d}\,(implicit)]$



Figura 2: Grafica de la superficie $z + 0.5\,x\,y + \frac{x^2}{3} = 0$

2.3. Funciones Vectoriales

Maxima nos permite trabajar con funciones vectoriales como graficar, parametrizar y realizar operaciones con ellas.

```
(%i1) load(draw);
      load(eigen);
      load(vect);
      draw3d(parametric(cos(t),cos(4*t),-sin(t),t, -4,4));
      r(t) := [cos(t), sin(t), t];
      float(r(1));
      limit(r(t),t,2);
      limit(r(t),t, 2, plus);
      limit(r(t), t,3,minus);
      define(rp(t), diff(r(t),t));
      float(rp(1));
      define( T(t), trigsimp( uvect( rp(t) ) ) );
      define(Tp(t), diff( T(t), t));
      define( N(t), trigsimp( uvect( Tp(t) ) ) );
      express(T(t)~N(t));
      define(B(t),trigsimp(%));
      float(B(1));
```

(%o1) $/usr/share/maxima/5.34.1/share/draw/draw.lisp$

(%o2) $/usr/share/maxima/5.34.1/share/matrix/eigen.mac$

(%o3) $/usr/share/maxima/5.34.1/share/vector/vect.mac$

(%o4) $[\mathrm{gr3d}\,(parametric)]$



Figura 3: Trayectoria descrita por $(cos(t), cos(4*t), -sin(t))$ donde $t \in [-4, 4]$

(%o5) $\mathrm{r}(t) := [\cos(t), \sin(t), t]$

(%o6) $[0.5403023058681398, 0.8414709848078965, 1.0]$

(%o7) $[\cos(2), \sin(2), 2]$

(%o8) $[\cos(2), \sin(2), 2]$

(%o9) $[\cos(3), \sin(3), 3]$

(%o10) $\mathrm{rp}(t) := [-\sin(t), \cos(t), 1]$

(%o11) $[-0.8414709848078965, 0.5403023058681398, 1.0]$

(%o12) $\mathrm{T}(t) := [-\frac{\sin(t)}{\sqrt{2}}, \frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$

(%o13) $\mathrm{Tp}(t) := [-\frac{\cos(t)}{\sqrt{2}}, -\frac{\sin(t)}{\sqrt{2}}, 0]$

(%o14) $\mathrm{N}(t) := [-\cos(t), -\sin(t), 0]$

(%o15) $[\frac{\sin(t)}{\sqrt{2}}, -\frac{\cos(t)}{\sqrt{2}}, \frac{\sin(t)^2}{\sqrt{2}} + \frac{\cos(t)^2}{\sqrt{2}}]$

(%o16) $\mathrm{B}(t) := [\frac{\sin(t)}{\sqrt{2}}, -\frac{\cos(t)}{\sqrt{2}}, \frac{1}{\sqrt{2}}]$

(%o17) $[0.5950098395293859, -0.3820514243700897, 0.7071067811865475]$

2.4. Longitud de Arco y Curvatura

En maxima nos permite realizar las operaciones para calcular estas cualidades de ecuaciones paramétricas.

```
(%i1) r(t) := [t, cos(t), sin(t)];
      rp(t) := [1,-sin(t), cos(t)];
      Tp(t) := [0,-cos(t), sin(t)]/sqrt(2);
      sqrt(Tp(t) . Tp(t))/sqrt(rp(t).rp(t));
      trigsimp(%);
      define(kappa(t),%);
      integrate(r(t),t);
      g(t) := [2*t,3*sin(t),3*cos(t)];
      define(gp(t) , diff(g(t),t));
      integrate(trigsimp(sqrt(gp(t).gp(t))),t,0,2*%pi);
      romberg(sqrt(gp(t).gp(t)),t,0,2*%pi);
```

(%o1) $\mathrm{r}\left(t\right):=\left[t,\cos\left(t\right),\sin\left(t\right)\right]$

(%o2) $\mathrm{rp}\left(t\right):=\left[1,-\sin\left(t\right),\cos\left(t\right)\right]$

(%o3) $\mathrm{Tp}\left(t\right):=\frac{\left[0,-\cos\left(t\right),\sin\left(t\right)\right]}{\sqrt{2}}$

(%o4) $\frac{\sqrt{\frac{\sin(t)^2}{2}+\frac{\cos(t)^2}{2}}}{\sqrt{\sin\left(t\right)^2+\cos\left(t\right)^2+1}}$

(%o5) $\frac{1}{2}$

(%o6) $\kappa\left(t\right):=\frac{1}{2}$

(%o7) $[\frac{t^2}{2},\sin\left(t\right),-\cos\left(t\right)]$

(%o8) $\mathrm{g}\left(t\right):=\left[2\,t,3\sin\left(t\right),3\cos\left(t\right)\right]$

(%o9) $\mathrm{gp}\left(t\right):=\left[2,3\cos\left(t\right),-3\sin\left(t\right)\right]$

(%o10) $2\sqrt{13}\,\pi$

(%o11) 22.65434679827795

3. Funciones de varias varibles

Maxima tiene la habilidad de trabajar con funciones de varias varibles, así como graficarlas.

```
(%i1) load(draw);
      f(x,y) := (5*x^2-2*y^2)^0.25;
      draw3d(explicit(f(x,y),x,-5,5,y,-5,5));
```

(%o1) $/usr/share/maxima/5.34.1/share/draw/draw.lisp$

(%o2) $\mathrm{f}\left(x,y\right):=\left(5\,x^2-2\,y^2\right)^{0.25}$

(%o3) $[\mathrm{gr3d}\left(explicit\right)]$



Figura 4: Superficie $f(x,y) = (5*x^2-2*y^2)^0.25$

```
(%i4)  draw3d(enhanced3d = true,
              explicit(f(x,y),x,-5,5,y,-5,5),
              palette = [green,blue,cyan,red]);
```

: (%o4) [gr3d ($explicit$)]



Figura 5: Superficie $f(x,y) = (5*x^2-2*y^2)^0.25$

```
(%i5)  draw3d(explicit(f(x,y),x,-5,5,y,-5,5),
            contour_levels = 20,
            contour        = map);
```

(%o5) [gr3d ($explicit$)]



Figura 6: curvas de nivel de $f(x,y) = (5 * x^2 - 2 * y^2)^0.25$

```
(%i6)  draw3d(enhanced3d = true,
              explicit(f(x,y),x,-5,5,y,-5,5),
              contour_levels = 20,
              contour = surface,
              palette = [green,blue,cyan,red],
              surface_hide = true);
```

(%o6) [gr3d ($explicit$)]



Figura 7: Superficie $f(x,y) = (5 * x^2 - 2 * y^2)^0.25$

3.1. Derivadas Parciales

Maxima nos permite realizar derivadas parciales

```
(%i1) diff(f(x,y),x,2,y,1,x,2);
      G : (1/32)*x^7*y^8;
      diff(G , x,2, y,1 , x,2);
```

(%o1) $\frac{d^5}{d\,x^4\,d\,y}\,\mathrm{f}\left(x,y\right)$

(%o2) $\frac{x^7\,y^8}{32}$

(%o3) $210\,x^3\,y^7$

3.2. Aproximación Lineal y Diferenciales

Con Maxima podemos realizar aproximaciones de funciones y ademas de poder expresar los diferenciales de las expresiones.

```
(%i1) f(x,y) := exp(x) * cos(y^2);
```

(%o1) $\mathrm{f}\left(x,y\right):=\exp\left(x\right)\,\cos\left(y^2\right)$

```
(%i2) taylor(f(x,y), [x,y],[1,2],1);
```

(%o2)/T/ $\cos\left(4\right)\,e+\left(\cos\left(4\right)\,e\,\left(x-1\right)-4\sin\left(4\right)\,e\,\left(y-2\right)\right)+...$

```
(%i3) diff(f(x,y));
```

(%o3) $e^x\cos\left(y^2\right)\,\mathrm{del}\left(x\right)-2\,e^x\,y\sin\left(y^2\right)\,\mathrm{del}\left(y\right)$

3.3. Regla de la cadena y derivación implicita

Se puede realizar la regla de la cadena y derivación implicita.

```
(%i1) f(x,y):= exp(x^3)*sin(4*y);
      [x,y] : [s^2*t, s*t^2];
```

(%o1) $\mathrm{f}\left(x,y\right):=\exp\left(x^3\right)\,\sin\left(4\,y\right)$

(%o2) $[s^2\,t,s\,t^2]$

```
(%i3) diff(f(x,y),s);
      diff(f(x,y),t);
```

(%o3) $6\,s^5\,t^3\,e^{s^6\,t^3}\sin\left(4\,s\,t^2\right)+4\,t^2\,e^{s^6\,t^3}\cos\left(4\,s\,t^2\right)$

(%o4) $3\,s^6\,t^2\,e^{s^6\,t^3}\,\sin\left(4\,s\,t^2\right)+8\,s\,t\,e^{s^6\,t^3}\,\cos\left(4\,s\,t^2\right)$

```
(%i5) diff(f(u,v),u);
      kill(x,y);
      diff(f(x,y),x);
```

(%o5) $3\,u^2\,e^{u^3}\,\sin\left(4\,v\right)$

(%o6) *done*

(%o7) $3\,x^2\,e^{x^3}\,\sin\left(4\,y\right)$

```
(%i8) F: 3*x*y^4*z^2 + 2*x*y*2*z-3*x*z-x;
      Fx: diff(F,x);
      Fy: diff(F,y);
      Fz: diff(F,z);
      [-Fx/Fy,-Fy/Fz];
```

(%o8) $3\,x\,y^4\,z^2+4\,x\,y\,z-3\,x\,z-x$

(%o9) $3\,y^4\,z^2+4\,y\,z-3\,z-1$

(%o10) $12\,x\,y^3\,z^2+4\,x\,z$

(%o11) $6\,x\,y^4\,z+4\,x\,y-3\,x$

(%o12) $\left[\frac{-3\,y^4\,z^2-4\,y\,z+3\,z+1}{12\,x\,y^3\,z^2+4\,x\,z},\frac{-12\,x\,y^3\,z^2-4\,x\,z}{6\,x\,y^4\,z+4\,x\,y-3\,x}\right]$

3.4. Derivada Direccional y Gradiente

En maxima es simple el calculo del gradiente, lo que nos permite calculos mas simples.

```
(%i1) load(vect);
      f(x,y):= exp(x^2)*sin(y);
      scalefactors([x,y]);
```

(%o1) */usr/share/maxima/*5.34.1*/share/vector/vect.mac*

(%o2) $\mathrm{f}\left(x,y\right):=\exp\left(x^2\right)\,\sin\left(y\right)$

(%o3) *done*

```
(%i4) gdf : grad(f(x,y));
      ev(express(gdf),diff);
      define(gdf(x,y),%);
```

(%o4) $\mathrm{grad}\left(e^{x^2}\,\sin\left(y\right)\right)$

(%o5) $\left[2\,x\,e^{x^2}\,\sin\left(y\right),e^{x^2}\,\cos\left(y\right)\right]$

(%o6) $\mathrm{gdf}\left(x,y\right):=\left[2\,x\,e^{x^2}\,\sin\left(y\right),e^{x^2}\,\cos\left(y\right)\right]$

```
(%i7)  v:[3,4];
       (gdf(1,2).v)/sqrt(v.v);
       ev(%,diff);
       float(%);
```

(%o7) $[3,4]$

(%o8) $\dfrac{6\,e\,\sin\left(2\right)+4\,e\,\cos\left(2\right)}{5}$

(%o9) $\dfrac{6\,e\,\sin\left(2\right)+4\,e\,\cos\left(2\right)}{5}$

(%o10) 2.061108499400332

```
(%i11) sqrt(gdf(1,2).gdf(1,2));
       float(ev(%,diff));
```

(%o11) $\sqrt{4\,e^2\,\sin\left(2\right)^2+e^2\,\cos\left(2\right)^2}$

(%o12) 5.071228088168654

3.5. Optimización y Extremos Locales

Con maxima podemos visualizar la grafica de la función y realizar los calculos para encontrar puntos de optimización.

```
(%i1)  load(draw);
       f(x,y) := x^3 +y^3-x*y;
       draw3d(enhanced3d =true,
              palette=[magenta, cyan, blue],
              explicit(f(x,y),x,-5,5,y,-5,5));
```

(%o1) $/usr/share/maxima/5.34.1/share/draw/draw.lisp$

(%o2) $\mathrm{f}\left(x,y\right):=x^3+y^3+\left(-x\right)\,y$

(%o3) $\left[\mathrm{gr3d}\left(explicit\right)\right]$



Figura 8: Superficie $f(x, y) = x^3 + y^3 + xy$

(%i4)
```
draw3d(contour_levels=10,
       contour=map,
       explicit(f(x,y),x,-5,5,y,-5,5));
```

(%o4) $[\mathrm{gr3d}\,(explicit)]$



Figura 9: Curvas de nivel de $f(x, y) = x^3 + y^3 + xy$

(%i5)
```
fx : diff(f(x,y),x);
fy : diff(f(x,y),y);
solve([fx,fy],[x,y]);
```

(%o5) $3\,x^2 - y$

(%o6) $3\,y^2 - x$

(%o7) $[[x=\frac{1}{3},y=\frac{1}{3}],[x=-\frac{\sqrt{3}\,i+1}{6},y=\frac{\sqrt{3}\,i-1}{6}],[x=\frac{\sqrt{3}\,i-1}{6},y=-\frac{\sqrt{3}\,i+1}{6}],[x=0,y=0]]$

```
(%i8) H: hessian(f(x,y),[x,y]);
      determinant(H);
```

(%o8) $\begin{pmatrix}6\,x & -1\\ -1 & 6\,y\end{pmatrix}$

(%o9) $36\,x\,y-1$

```
(%i10) subst([x=1/3, y=1/3],diff(fx,x));
       subst([x=1/3, y=1/3],determinant(H));
       f(1/3,1/3);
```

(%o10) 2

(%o11) 3

(%o12) $-\frac{1}{27}$

```
(%i13) subst([x=0, y=0],diff(fx,x));
       subst([x=0, y=0],determinant(H));
       f(0,0);
```

(%o13) 0

(%o14) -1

(%o15) 0

3.6. Multiplicadores de Lagrange

```
(%i1) f(x,y) := x^2+2*y^2;
      g : y^2+x^2;
```

(%o1) $\mathrm{f}\left(x,y\right):=x^2+2\,y^2$

(%o2) y^2+x^2

```
(%i3) eq1: diff(f(x,y),x)=h*diff(g,x);
      eq2: diff(f(x,y),y)=h*diff(g,y);
      eq3: g=1;
```

(%o3) $2\,x=2\,h\,x$

(%o4) $4\,y=2\,h\,y$

(%o5) $y^2+x^2=1$

(%i6) `solve([eq1,eq2,eq3],[x,y,h]);`

(%o6) $[[x=1,y=0,h=1],[x=-1,y=0,h=1],[x=0,y=-1,h=2],[x=0,y=1,h=2]]$

(%i7) `[f(1,0),f(-1,0),f(0,-1),f(0,1)];`

(%o7) $[1,1,2,2]$

4. Integración Multiple

Maxima nos permite realizar integrales de varias variables.

4.1. Integrales Dobles

Con el comando `integrate()` nos permite realizar las integrales .

(%i1) `f: 4*x^3-4*x*y;`

(%o1) $4\,x^3-4\,x\,y$

(%i2) `integrate(integrate(f,y),x);`

(%o2) $x^4\,y-x^2\,y^2$

(%i3) `integrate(integrate(f,y,x^1/2,2-x),x,0,1);`

(%o3) $-\frac{109}{120}$

4.2. Coordenadas Polares

Con maxima se puede cambiar a coordenadas polares para facilitar el calculo.

(%i1) `f(x,y):= 4*x^2+4*y^2;`

(%o1) $\mathrm{f}\left(x,y\right):=4\,x^2+4\,y^2$

(%i2) `[x,y]: [r*cos(theta), r*sin(theta)];`

(%o2) $\left[r\cos\left(\theta\right),r\sin\left(\theta\right)\right]$

(%i3) `integrate(integrate(f(x,y)*r,r,0,2*cos(theta)),theta,-%pi/2,%pi/2);`

(%o3) $6\,\pi$

4.3. Integrales Triples

Tambien es posible realizar integrales de 3 parametros.

```
(%i1) f(x,y,z):= x^2*y*z;
      integrate(integrate(integrate(f(x,y,z),z,0,x+y),y,0,-x),x,0,1);
```

(%o1) $\mathrm{f}(x, y, z) := x^2\, y\, z$

(%o2) $\frac{1}{168}$

4.4. Integreles en Coordenadas Cilindricas y Esfericas

Con maxima se puede cambiar a otras coordenadas para facilitar el calculo.

```
(%i1) f(x,y,z):= y*z;
      [x,y,z]:[r*cos(theta),r*sin(theta),z];
      integrate(integrate(integrate(f(x,y,z)*r,z,0,3),r,0,2),theta,0,%pi);
      kill(f,x,y,z);
```

(%o1) $\mathrm{f}(x, y, z) := y\, z$

(%o2) $[r\cos(\theta), r\sin(\theta), z]$

(%o3) 24

(%o4) *done*

```
(%i5) f(x,y,z):= x*z;
      [x,y,z]: [rho*sin(phi)*cos(theta),rho*sin(phi)*sin(theta),rho*cos(phi)];
      integrate(integrate(integrate(f(x,y,z)*rho^2*sin(phi),rho,0,1),theta,0,%pi),
      kill(f,x,y,z);
```

(%o5) $\mathrm{f}(x, y, z) := x\, z$

(%o6) $[\sin(\phi)\ \rho\cos(\theta), \sin(\phi)\ \rho\sin(\theta), \cos(\phi)\ \rho]$

(%o7) 0

(%o8) *done*

4.5. Cambio de variable

Con maxima se puede realizar cambio variable para facilitar el calculo.

```
(%i1) f(x,y) := x+y;
      [x,y]: [u^3-v^4, 5*u*v];
```

(%o1) $\mathrm{f}(x, y) := x + y$

(%o2) $[u^3 - v^4, 5\, u\, v]$

```
(%i3)  J: jacobian([x,y],[u,v]);
       J: determinant(J);
```

$$(\%o3)\quad \begin{pmatrix} 3\,u^2 & -4\,v^3 \\ 5\,v & 5\,u \end{pmatrix}$$

$$(\%o4)\quad 20\,v^4+15\,u^3$$

```
(%i5)  integrate(integrate(f(x,y)*J,u,1,2),v,3,4);
```

$$(\%o5)\quad -\frac{113349305}{252}$$

5. Cálculo Vectorial

Maxima nos permite trabajar con campos vectoriales.

5.1. Campo Vectorial

Maxima nos permite graficar los campos vectoriales

```
(%i1)  load(draw);
       load(vect);
       F(x,y):= (x^2,y^2);
```

$(\%o1)\quad /usr/share/maxima/5.34.1/share/draw/draw.lisp$

$(\%o2)\quad /usr/share/maxima/5.34.1/share/vector/vect.mac$

$(\%o3)\quad \mathrm{F}\left(x,y\right):=\left(x^2,y^2\right)$

Campo Vectorial Dos Dimensiones: Graficación del campo vectorial.

```
(%i4)  coord: setify(makelist(k,k,-6,6));
       points2d :listify(cartesian_product(coord,coord));
       vf2d(x,y):= vector([x,y],[4*cos(y),x^2]/10);
       vect2: makelist(vf2d(k[1],k[2]), k, points2d);
       apply ( draw2d , append ([head_length=0.2], [color = green ] , vect2 ));
```



Figura 10: Campo $f(x,y) = (4\cos(y), x^2)$

Campo Gradiente: Graficación del Campo Gradiente.

```
(%i9)  kill(f,x,y,gdf);
       f(x,y) := cos(x^2) - y^2;
       scalefactors ([ x , y ]);
       gdf(x,y):= grad(f(x,y));
       ev(express(gdf(x,y)),diff);
       define(gdf(x,y),%);

(%i15) coord: setify(makelist(k,k,-6,6));
       points2d : listify(cartesian_product(coord,coord));
       vf2d(x,y):= vector([x,y],gdf(x,y)/10);
       vect2: makelist(vf2d(k[1],k[2]),k, points2d);
       apply(draw2d, append([head_length=0.25, color=green], vect2));
```



Figura 11: Campo gradiente $f(x, y) = (-2x\sin(x^2), -2y)$

Campo Vectorial Tres Dimensiones: Graficación Campo vectorial 3 dimensiones.

```
(%i20) coord: setify(makelist(k,k,-3,3));
       points3d : listify(cartesian_product(coord, coord, coord));
       vf3d(x,y,z):= vector([x,y,z],[z,x*z,y]/8);
       vect3 : makelist(vf3d(k[1],k[2],k[3]),k,points3d);
       apply(draw3d, append([color=red,head_length=0.1],vect3));
```



Figura 12: Campo $f(x, y, z) = (z, xz, y)$

5.2. Integral de Linea

Con maxima es simple realizar una integral de linea.

```
(%i1) f(x,y) := x^2+y^2;
      [x,y] : [cos(t), sin(2*t)];
      rp: diff([x,y],t);
      romberg(f(x,y)*sqrt(rp.rp), t, 0,1);
```

(%o1) $f(x, y) := x^2 + y^2$

(%o2) $[\cos(t), \sin(2t)]$

(%o3) $[-\sin(t), 2\cos(2t)]$

(%o4) 1.635879048260742

```
(%i5) F(x,y,z) := [-x*y^3, x*z, y*z^2];
      [x,y,z]: [t^2,t^3,t^4];
      romberg(F(x,y,z).diff([x,y,z],t),t,0,1);
```

(%o5) $F(x, y, z) := [(-x)\, y^3, x\, z, y\, z^2]$

(%o6) $[t^2, t^3, t^4]$

(%o7) 0.4461538461603604

5.3. Campos Conservativos y Encontrando Potenciales Escalares

Con la función `curl()` podemos ver si los campos son conservativos, y podemos encontrar el potencial escalar con la función `potential()`.

```
(%i1) load(vect);
      F(x,y):= [4*x^3-5*y^2,5*y^3-3*x];
      scalefactors([x,y]);
```

(%o1) $/usr/share/maxima/5.34.1/share/vector/vect.mac$

(%o2) $F(x, y) := [4x^3 - 5y^2, 5y^3 - 3x]$

(%o3) *done*

```
(%i4) curl(F(x,y));
      express(%);
      ev(%,diff);
```

(%o4) $\mathrm{curl}\left([4x^3 - 5y^2, 5y^3 - 3x]\right)$

(%o5) $\frac{d}{d\,x}\left(5y^3 - 3x\right) - \frac{d}{d\,y}\left(4x^3 - 5y^2\right)$

(%o6) $10\,y-3$

```
(%i7) F(x,y):= [x^3+5*y,5*y^3+5*x];
      ev(express(curl(F(x,y))),diff);
```

(%o7) $\mathrm{F}\left(x,y\right):=[x^3+5\,y,5\,y^3+5\,x]$

(%o8) 0

```
(%i9) F(u,v):= [u^3+5*v,5*v^3+5*u];
      scalefactors([u,v]);
      potential(F(u,v));
      define(f(u,v),%);
      f(2,3)-f(0,1);
```

(%o9) $\mathrm{F}\left(u,v\right):=[u^3+5\,v,5\,v^3+5\,u]$

(%o10) *done*

(%o11) $\dfrac{5\,v^4+20\,u\,v+u^4}{4}$

(%o12) $\mathrm{f}\left(u,v\right):=\dfrac{5\,v^4+20\,u\,v+u^4}{4}$

(%o13) 134

Referencias

[1] Source Forge, *Maxima*. Recuperado en abril de 2016 de `http://maxima.sourceforge.net/es/`

[2] Kerns, Jay. *Multivariable Calculus with Maxima*. Recuperado en abril de 2016 de `http://gkerns.people.ysu.edu/maxima/maximaintro/`

[3] Lizárraga, C. *Actividad 8 (2016-1)*. Recuperado en abril de 2016 de `http://computacional1.pbworks.com/w/page/106192917/Actividad%208%20(2016-1)`